Compound Options - 08-20-16

N. T. Gladd

Initialization: Be sure the files *NTGStylesheet2.nb* and *NTGUtilityFunctions.m* is are in the same directory as that from which this notebook was loaded. Then execute the cell immediately below by mousing left on the cell bar to the right of that cell and then typing “shift” + “enter”. Respond “Yes” in response to the query to evaluate initialization cells.

In[15]:=
```
SetDirectory[NotebookDirectory[]];
(* set directory where source files are located *)
SetOptions[EvaluationNotebook[], (* load the StyleSheet *)
 StyleDefinitions → Get["NTGStylesheet2.nb"]];
Get["NTGUtilityFunctions.m"]; (* Load utilities package *)
```

Original notes *Compound Options (Kaminski)*, *Compound Option Derivation 03-05-08,* Compound Option Derivation 09-04-10

Purpose

Original notes *Compound Options (Kaminski)*, *Compound Option Derivation 03-05-08,* Compound Option Derivation 09-04-10

Compound options — options on options — are a relatively simple form of exotic options first treated in the literature by Geske in 1978. http://citeseerx.ist.psu.edu/viewdoc/download?-doi=10.1.1.453.2801&rep=rep1&type=pdf

Compound options are not traded on exchanges but are available to institutional traders through the over-the-counter equity derivative trading desks of large investment banks. Compound options are sometimes useful in non trading contexts — a web search indicates they have developed a following in the “real options” subgenre of decision science. They are also used when valuing employee stock options, complex instruments for which an active market does not exist.

I use *Mathematica* to derive numerical forms and a semi-closed form for the fair value of a representa-tive compound option. The calculations are somewhat long and complicated and facilitated by the use computer symbolic manipulation. As well, some of the techniques invoked at various stages of the calculation are useful in other contexts.

Background

Consider a option struck at K_O at time t on a stock option struck at K_S at time T_O and expiring at time T_S



As an archetype of compound options, consider a call option struck on a call option on a stock S.

$$\text{COC}(t) = e^{-r(T_O - t)}\, \mathbb{E}_Q[\max[C[S(T_O),\ K_O,\ T_S - T_O] - K_O,\ 0]] \tag{1}$$

where

$$C[S(T_O),\ K_O,\ T_S - T_O] = e^{-r(T_S - T_O)}\, \mathbb{E}_Q[\max[S(T_S) - K_S,\ 0]] \tag{2}$$

is a standard (so-called plain vanilla) call option. Here, $\mathbb{E}_Q[...]$ denotes expectation taken under the risk neutral probability measure. Famously, under geometric Brownian motion stock dynamics, the standard call option is fairly valued by the Black-Scholes formula

$$C[S,\ K,\ r,\ q,\ \sigma,\ \tau] = e^{-q\,\tau}\, S\, \mathcal{N}(d) - e^{-r\,\tau}\, K\, \mathcal{N}\left(d - \sigma\sqrt{\tau}\right) \tag{3}$$

$$d = \frac{\log(S/K) + \left(r - q + \frac{\sigma^2}{2}\right)\tau}{\sigma\sqrt{\tau}}$$

$$\mathcal{N}(x) = \int_{-\infty}^{x} n(t)\, dt \qquad \text{cumulative standard normal distribution} \tag{4}$$

$$n(t) = \frac{1}{\sqrt{2\,\pi}}\, e^{-\frac{t^2}{2}} \qquad \text{standard normal distribution}$$

In the context of equations (1) and (2), S = $S(T_O)$, K = K_O, $\tau = T_S - T_O$.
These distributions are readily available in Mathematica.

In[17]:=
```
Module[{dist, info},
 dist = NormalDistribution[];
 info = {{n[x], PDF[dist, x]}, {N[x], CDF[dist, x]}};
 LGrid[info, "Normal distributions"]]
```

Out[17]=

Normal distributions

$n[x]$	$\frac{e^{-\frac{x^2}{2}}}{\sqrt{2\pi}}$
$\mathcal{N}[x]$	$\frac{1}{2}\text{Erfc}\left[-\frac{x}{\sqrt{2}}\right]$

The stock dynamics under GBM are

$$S(T,\ \epsilon) = S(t)\exp\left[\left(r - q - \frac{\sigma^2}{2}\right)(T - t) + \sigma\,\epsilon\sqrt{(T - t)}\right] \tag{5}$$

where ϵ is a random number chosen from $n(\epsilon)$.

Equation (1) can be expressed as an integral over ϵ

$$\mathrm{COC}(t) = e^{-r(T_O-t)} \int_{-\infty}^{\infty} n(\epsilon) \max[C[S(T_O, \epsilon), K_O, T_S - T_O] - K_O, 0] \, d\epsilon$$
$$= e^{-r(T_O-t)} \int_{-\infty}^{\infty} n(\epsilon) \max\Big(\Big(e^{-q\tau} S(T_O, \epsilon) \mathcal{N}(d(\epsilon)) - e^{-r\tau} K_S \mathcal{N}\Big(d(\epsilon) - \sigma\sqrt{\tau}\Big)\Big) - K_O\Big) d\epsilon \quad (6)$$

where I have made it clear that the stock price at time T_O depends on the random variable ϵ.

If interest is limited to obtaining a numerical answer, equation (6) or equation (7) can be immediately valued.

However, Geske (reference above) calculated an explicit expression for (6) in terms of the cumulative binormal distribution. Below, I use the pattern recognition capabilities of Mathematica to derive the Geske result. Specifically, from (expression w1[16] below) I have

$$\mathrm{COC}(t=0) = e^{-\frac{b^2}{2}} S_0 \mathcal{I}(a, b, b, \epsilon_c) - K_S \mathcal{I}\Big(a - \sigma\sqrt{\tau}, b, 0, \epsilon_c\Big) - K_O \mathcal{N}(-\epsilon_c) \quad (7)$$

where

$$\tau = -T_O + T_S$$
$$b = \sigma\sqrt{T_O}$$
$$a = -\frac{b^2}{2\sigma\sqrt{\tau}} + \frac{\sigma\sqrt{\tau}}{2} - \frac{\mathrm{Log}[K_S]}{\sigma\sqrt{\tau}} + \frac{\mathrm{Log}[S_0]}{\sigma\sqrt{\tau}} \quad (8)$$
$$b = \frac{b}{\sigma\sqrt{\tau}}$$
$$S(t=0) = S_0$$

and the quantity ϵ_c is the solution of

$$C[S(T_O, \epsilon_c), K_O, T_S - T_O] = K_O \quad (9)$$

which, for general parameters, must be obtained numerically.

Also, as calculated in Appendix A.

$$\mathcal{I}(a, b, c, d) =$$
$$\int_d^{\infty} dx \, e^{a x} n(x) \mathcal{N}(b x + c) = == e^{\frac{c^2}{2}} \left(\mathcal{N}\Big[\frac{a + b c}{\sqrt{1+b^2}}\Big] - \mathcal{N}_2\Big[-c + d, \frac{a + b c}{\sqrt{1+b^2}}, -\frac{b}{\sqrt{1+b^2}}\Big]\right) \quad (10)$$

with $\mathcal{N}_2$ being the standard cumulative binormal distribution

$$\mathcal{N}_2(a, b, \rho) = \int_{-\infty}^{a} dx \int_{-\infty}^{b} dy \, n_2(x, y, \rho)$$

$$n_2(x,\ y,\ \rho) = \frac{e^{-\frac{x^2+y^2-2\,x\,y\,\rho}{2\,(1-\rho^2)}}}{2\,\pi\sqrt{1-\rho^2}}$$

I have performed the calculation for the special case r = q = t = 0 because of the bulkiness of intermediate expressions. However, the calculation is essentially algorithmic and could be repeated without the assumption of zero rates.

The valuation formula (7) should be considered to be semi-closed. For one thing the parameter ϵ_c must be obtained numerically. Secondly, there is no readily available closed form for $\mathcal{N}_2$ for general parameters.

In[18]:=
```
{PDF[BinormalDistribution[ρ], {x, y}], CDF[BinormalDistribution[ρ], {x, y}]}
```

Out[18]=

$$\left\{\frac{e^{-\frac{x^2+y^2-2\,x\,y\,\rho}{2\,(1-\rho^2)}}}{2\,\pi\,\sqrt{1-\rho^2}},\ \text{CDF}[\text{BinormalDistribution}[\rho],\ \{x,\ y\}]\right\}$$

Guide to calculations below

Section 1 Equation (7) is derived.
Section 2 An explicit form for $\mathcal{I}$(a, b, c, d).
Section 3 Comparative numerical calculations are performed
Appendix A Some identities involving $\mathcal{N}_2$(a, b, ρ) are derived. The method used involves nontrivial multiple definitions of the evaluation function that streamline intermediate calculations.
Appendix B The figure above is constructed.

1 Derivation of closed form for COC

$$\text{COC}(t) = e^{-r(T_O-t)}\int_{-\infty}^{\infty} n(\epsilon)\max[C[S(T_O,\ \epsilon),\ K_O,\ T_S-T_O]-K_O,\ 0]\,d\epsilon$$

$$= e^{-r(T_O-t)}\int_{-\infty}^{\infty} n(\epsilon)\max\left(\left(e^{-q\,\tau}\,S(T_O,\ \epsilon)\,\mathcal{N}(d(\epsilon)) - e^{-r\,\tau}\,K_S\,\mathcal{N}\left(d(\epsilon)-\sigma\sqrt{\tau}\right)\right)-K_O\right)d\epsilon$$

The objective is to reexpress equation (6) is terms of known functions. Note first that the Max term has the effect of introducing a cutoff for the integration range.

In[19]:=
```
Module[{St = 100, r = 0.01, q = 0.03, σ = 0.2,
  TO = 1/12, KS = 100, TS = 3/12, τ, KO, root, inset, SO, d, n, N},
 τ = TS - TO;
 KO = EuroCall[St, KS, r, q, σ, TS - TO];
 SO[ϵ_] := St Exp[(r - q + σ^2/2) TO + ϵ σ TO];
 d[ϵ_] = 1/(σ Sqrt[τ]) (Log[SO[ϵ]/KS] + (r - q + σ^2/2) τ);
 n[ϵ_] := PDF[NormalDistribution[], ϵ];
 N[a_] := CDF[NormalDistribution[], a];
 root =
  FindRoot[Exp[-q τ] SO[ϵ] N[d[ϵ]] - Exp[-q τ] KS N[-σ Sqrt[τ] + d[ϵ]] - KO == 0, {ϵ, 0}];
 inset = Stl@StringForm["ϵc = ``", root[[1, 2]]];
 Plot[n[ϵ] Max[Exp[-q τ] SO[ϵ] N[d[ϵ]] - Exp[-q τ] KS N[-σ Sqrt[τ] + d[ϵ]] - KO, 0],
  {ϵ, -5, 5}, PlotRange → All,
  Epilog → {Inset[inset, {-2.5, 0.1}]}, AxesLabel → {Stl["ϵ"], ""},
  PlotLabel → Stl["integrand of expression for COC"], PlotStyle → Black]]
```

Out[19]=



The cutoff value ϵ_c is the solution of

$$e^{-q\,\tau}\,S(T_O,\ \epsilon_c)\,\mathcal{N}(d(\epsilon_c)) - e^{-r\,\tau}\,K_S\,\mathcal{N}\left(d(\epsilon_c) - \sigma\sqrt{\tau}\right) - K_O = 0 \tag{12}$$

and, for general parameters, has to be determined numerically.

Given a value for ϵ_c, the integral to be evaluated is

$$\text{COC}(t) = e^{-r(T_O-t)}\int_{\epsilon_c}^{\infty} n(\epsilon)\left(e^{-q\,\tau}\,S(t)\exp\left[\left(r - q - \frac{\sigma^2}{2}\right)(T_O-t) + \sigma\,\epsilon\sqrt{(T_O-t)}\right]\mathcal{N}(d(\epsilon)) - e^{-r\,\tau}\,K_S\,\mathcal{N}\left(d(\epsilon) - \sigma\sqrt{\tau}\right) - K_O\right)d\epsilon \tag{13}$$

and it turns out that integrals involving products of e^{ϵ}, n(ϵ), and $\mathcal{N}$(a ϵ + b) have closed forms involving the cumulative binormal distribution.

So — I use Mathematica to manipulate the integral (8) into a form where the pattern of the integrand in (9) can be matched to a known integral. Rather than writing down the integral (8) I start with the basic definitions of COC, C, d, and S(T) and construct the required integral

From above the definition of a call on call compound option (COC) is

In[20]:=
```
w1[1] = COC == Exp[-r (T0 - t)] Int[n[ϵ] (C[S[T0], KS, TS - T0] - K0), {ϵ, ϵc, ∞}]
```

Out[20]=

$$COC == e^{-r(-t+T_0)}\,\mathrm{Int}\left[\left(C[S[T_0], K_S, -T_0+T_S] - K_0\right) n[\epsilon], \{\epsilon, \epsilon_c, \infty\}\right]$$

In writing this expression, I purposely use the artificial structure Int[integrand, limits] rather than Mathematica's Integrate[integrand, limits] function, which would immediately attempt to evaluate. The Int structure permits symbolic manipulations.

The call option is valued by the Black-Scholes formula

In[21]:=
```
w1[2] = C[S[T0], KS, TS - T0] ==
  Exp[-q (TS - T0)] S[T0] N[d] - Exp[-r (TS - T0)] KS N[d - σ Sqrt[TS - T0]]
```

Out[21]=

$$C[S[T_0], K_S, -T_0+T_S] == e^{-q(-T_0+T_S)} S[T_0]\,\mathcal{N}[d] - e^{-r(-T_0+T_S)} K_S\,\mathcal{N}\left[d - \sigma\sqrt{-T_0+T_S}\right]$$

where

In[22]:=
```
w1[3] = d == (Log[S[T0] / KS] + (r - q + σ^2/2) (TS - T0)) / (σ Sqrt[(TS - T0)])
```

Out[22]=

$$d == \left(\mathrm{Log}\left[\frac{S[T_0]}{K_S}\right] + \left(-q+r+\frac{\sigma^2}{2}\right)(-T_0+T_S)\right) \Big/ \left(\sigma\sqrt{-T_0+T_S}\right)$$

The GBM stock dynamics are

In[23]:=
```
w1[4] = S[T0] == S[t] Exp[(r - q - σ^2/2) (T0 - t) + σ ϵ Sqrt[T0 - t]]
```

Out[23]=

$$S[T_0] == e^{\epsilon\sigma\sqrt{-t+T_0} + \left(-q+r-\frac{\sigma^2}{2}\right)(-t+T_0)} S[t]$$

with ϵ being a random number chosen from $n(\epsilon)$.

Clearly, the expressions to be manipulated will be unwieldy. For the purpose of identifying the individual integral expressions, the simplifications r = q = t = 0 can be made. Once the collection of of operations leading to the final result has been established, it is trivial to repeat the calculation with symbolic values for r, q, and t. I also introduce some simplifying parameter substitutions

In[24]:=
```
simplifyingSubs = {-T0 + TS → τ, σ Sqrt[T0] → b, σ^2 T0 → b^2, S[0] → S0}
```

Out[24]=

$$\{-T_0+T_S \to \tau, \sigma\sqrt{T_0} \to b, \sigma^2 T_0 \to b^2, S[0] \to S_0\}$$

Impose the simplifications

In[25]:=
```
w1[5] = {w1[1], w1[2], w1[3], w1[4]} /. {r → 0, q → 0, t → 0} /. simplifyingSubs
```

Out[25]=

$$\left\{COC == Int\left[\left(C[S[T_0], K_S, \tau] - K_0\right) n[\epsilon], \{\epsilon, \epsilon_c, \infty\}\right],\ C[S[T_0], K_S, \tau] == S[T_0]\, \mathcal{N}[d] - K_S\, \mathcal{N}\left[d - \sigma\sqrt{\tau}\right],\ d == \frac{\frac{\sigma^2 \tau}{2} + Log\left[\frac{S[T_0]}{K_S}\right]}{\sigma\sqrt{\tau}},\ S[T_0] == e^{-\frac{b^2}{2} + b\,\epsilon} S_0\right\}$$

The combined form for the COC integral is

In[26]:=
```
w1[6] = w1[5][[1]] /. (w1[5][[2]] // ER) /. (w1[5][[4]] // ER)
```

Out[26]=

$$COC == Int\left[n[\epsilon]\left(-K_0 + e^{-\frac{b^2}{2} + b\,\epsilon} S_0\, \mathcal{N}[d] - K_S\, \mathcal{N}\left[d - \sigma\sqrt{\tau}\right]\right), \{\epsilon, \epsilon_c, \infty\}\right]$$

A first step is to determine the explicit dependence of the argument of $\mathcal{N}$(d) on ϵ

In[27]:=
```
w1[7] = w1[5][[3]] /. (w1[5][[4]] // ER)
```

Out[27]=

$$d == \frac{\frac{\sigma^2 \tau}{2} + Log\left[\frac{e^{-\frac{b^2}{2} + b\,\epsilon} S_0}{K_S}\right]}{\sigma\sqrt{\tau}}$$

In[28]:=
```
w1[8] = w1[7] /. Log[a_] :→ PowerExpand@Log[a]
```

Out[28]=

$$d == \frac{1}{\sigma\sqrt{\tau}}\left(-\frac{b^2}{2} + b\,\epsilon + \frac{\sigma^2 \tau}{2} - Log[K_S] + Log[S_0]\right)$$

I note that this has the form $a + b\,\epsilon$ and identify specific values for a and b

In[29]:=
```
w1[9] = {a, b} → (ExpandAll@w1[8][[2]] /. a_ + b_ϵ /; FreeQ[b1, ϵ, ∞] → {a, b})
```

Out[29]=

$$\{a, b\} \to \left\{-\frac{b^2}{2\sigma\sqrt{\tau}} + \frac{\sigma\sqrt{\tau}}{2} - \frac{Log[K_S]}{\sigma\sqrt{\tau}} + \frac{Log[S_0]}{\sigma\sqrt{\tau}},\ \frac{b}{\sigma\sqrt{\tau}}\right\}$$

With knowledge of explicit forms for a and b, I can simplify the main expression

In[30]:=
```
w1[10] = w1[5][[1]] /. (w1[5][[2]] // ER) /. (w1[5][[4]] // ER)
```

Out[30]=

$$COC == Int\left[n[\epsilon]\left(-K_0 + e^{-\frac{b^2}{2} + b\,\epsilon} S_0\, \mathcal{N}[d] - K_S\, \mathcal{N}\left[d - \sigma\sqrt{\tau}\right]\right), \{\epsilon, \epsilon_c, \infty\}\right]$$

In[31]:=
```
w1[11] = w1[10] /. d → a + b ϵ
```

Out[31]=

$$COC == Int\left[n[\epsilon]\left(-K_0 + e^{-\frac{b^2}{2} + b\,\epsilon} S_0\, \mathcal{N}[a + b\,\epsilon] - K_S\, \mathcal{N}\left[a + b\,\epsilon - \sigma\sqrt{\tau}\right]\right), \{\epsilon, \epsilon_c, \infty\}\right]$$

Next, some rules are applied that distribute the operator Int over summation, and move constant terms outside the Int structure

In[32]:=
```
w1[12] = ExpandAll@w1[11] //. Int[a_ + b_, c_] → Int[a, c] + Int[b, c] //.
   Int[a_ b_, c_] /; FreeQ[a, ϵ, ∞] → a Int[b, c] /.
  Int[c_ Exp[a_ + b_ ϵ], lim_] → Exp[a] Int[c Exp[b ϵ], lim]
```

Out[32]=

$$COC == -Int[n[\epsilon], \{\epsilon, \epsilon_c, \infty\}] K_0 - Int\left[n[\epsilon] \mathcal{N}\left[a + b\epsilon - \sigma\sqrt{\tau}\right], \{\epsilon, \epsilon_c, \infty\}\right] K_S + e^{-\frac{b^2}{2}} Int\left[e^{b\epsilon} n[\epsilon] \mathcal{N}[a + b\epsilon], \{\epsilon, \epsilon_c, \infty\}\right] S_0$$

The first term can be evaluated immediately.

In[33]:=
```
ErfRules = {Erfc[x_] → 1 - Erf[x], Erf[x_] → 2 𝒩[√2 x] - 1};
```

In[34]:=
```
w1[13] = w1[12][[2, 1]] →
  (w1[12][[2, 1]] /. n[ϵ] → PDF[NormalDistribution[], ϵ] /. Int → Integrate //.
     ErfRules /. 𝒩[a_] → 1 - 𝒩[-a] // ExpandAll)
```

Out[34]=

$$-Int[n[\epsilon], \{\epsilon, \epsilon_c, \infty\}] K_0 \to -K_0 \mathcal{N}[-\epsilon_c]$$

Mathematica evaluates such integrals in terms of Erf. I use ErfRules to transform to the cumulative normal distribution function that is preferred in the financial literature.

In[35]:=
```
w1[14] = w1[12] /. w1[13]
```

Out[35]=

$$COC == -Int\left[n[\epsilon] \mathcal{N}\left[a + b\epsilon - \sigma\sqrt{\tau}\right], \{\epsilon, \epsilon_c, \infty\}\right] K_S + e^{-\frac{b^2}{2}} Int\left[e^{b\epsilon} n[\epsilon] \mathcal{N}[a + b\epsilon], \{\epsilon, \epsilon_c, \infty\}\right] S_0 - K_0 \mathcal{N}[-\epsilon_c]$$

I next introduce pattern matching rules for the remaining integrals, which have the structure

$$\mathcal{I}(a, b, c, d) \equiv \int_d^\infty d\epsilon\, e^{c\,\epsilon}\, n(\epsilon)\, \mathcal{N}(a + b\,\epsilon)$$

and will be evaluated in Section 2.

In[36]:=
```
w1[15] = w1[14] /. Int[e^(c_ ϵ) n[ϵ] 𝒩[a_ + b_ ϵ], {ϵ, d_, ∞}] /;
    And[FreeQ[b, ϵ, ∞], FreeQ[c, ϵ, ∞]] → 𝓘[a, b, c, d]
```

Out[36]=

$$COC == -Int\left[n[\epsilon] \mathcal{N}\left[a + b\epsilon - \sigma\sqrt{\tau}\right], \{\epsilon, \epsilon_c, \infty\}\right] K_S + e^{-\frac{b^2}{2}} S_0 \mathcal{I}[a, b, b, \epsilon_c] - K_0 \mathcal{N}[-\epsilon_c]$$

In[37]:=
```
w1[16] =
 w1[15] /. Int[n[ϵ] 𝒩[a_ + b_ ϵ], {ϵ, d_, ∞}] /; FreeQ[b, ϵ, ∞] → 𝓘[a, b, 0, d]
```

Out[37]=

$$COC == e^{-\frac{b^2}{2}} S_0 \mathcal{I}[a, b, b, \epsilon_c] - K_S \mathcal{I}\left[a - \sigma\sqrt{\tau}, b, 0, \epsilon_c\right] - K_0 \mathcal{N}[-\epsilon_c]$$

Thus, the COC has been expressed in terms of an integral $\mathcal{I}$(a,b,c,d) which will be shown to have a closed form. Actually, I refer to it as semi-closed since the cutoff parameter ϵ_c must be determined numerically. Also, unless one has confidence in an approximate form of the cumulative standard binormal distribution, $\mathcal{N}_2$ must also be evaluated numerically.

I collect results and construct a function that can be used for numerical studies. I avoid the use of subscripts within Function definitions. For some of the reasons why, see http://mathematica.stackexchange.com/questions/18393/what-are-the-most-common-pitfalls-awaiting-new-users/18395#18395

For τ and b

In[38]:=
```
(simplifyingSubs [[1]] // Reverse) /. {T0 → TO, TS → TS}
```
Out[38]=
$$\tau \to -\mathrm{TO} + \mathrm{TS}$$

In[39]:=
```
(simplifyingSubs [[2]] // Reverse) /. {T0 → TO}
```
Out[39]=
$$\mathrm{b} \to \sqrt{\mathrm{TO}}\ \sigma$$

For a and b

In[40]:=
```
w1[9] /. {S0 → St, KS → KS}
```
Out[40]=
$$\{a, b\} \to \left\{-\frac{\mathrm{b}^2}{2\,\sigma\sqrt{\tau}} + \frac{\sigma\sqrt{\tau}}{2} - \frac{\mathrm{Log}[\mathrm{KS}]}{\sigma\sqrt{\tau}} + \frac{\mathrm{Log}[\mathrm{St}]}{\sigma\sqrt{\tau}},\ \frac{\mathrm{b}}{\sigma\sqrt{\tau}}\right\}$$

From Section 2

In[41]:=
$$\text{w2["final result"]} = \mathcal{I}[\mathrm{a, b, c, d}] = e^{\frac{c^2}{2}}\left(\mathcal{N}\left[\frac{a+b\,c}{\sqrt{1+b^2}}\right] - \mathcal{N}_2\left[-c+d,\ \frac{a+b\,c}{\sqrt{1+b^2}},\ -\frac{b}{\sqrt{1+b^2}}\right]\right)$$

Out[41]=
$$\mathcal{I}[\mathrm{a, b, c, d}] = e^{\frac{c^2}{2}}\left(\mathcal{N}\left[\frac{a+b\,c}{\sqrt{1+b^2}}\right] - \mathcal{N}_2\left[-c+d,\ \frac{a+b\,c}{\sqrt{1+b^2}},\ -\frac{b}{\sqrt{1+b^2}}\right]\right)$$

In preparation for numerical calculations

In[42]:=
```
w2["final result"] /. 𝒩[a_] → 𝒩Numerical[a] /.
 𝒩2[a_, b_, ρ_] → 𝒩2Numerical[a, b, ρ]
```
Out[42]=
$$\mathcal{I}[\mathrm{a, b, c, d}] = e^{\frac{c^2}{2}}\left(\frac{1}{2}\,\mathrm{Erfc}\left[-\frac{a+b\,c}{\sqrt{2}\,\sqrt{1+b^2}}\right] - \mathcal{N}\mathrm{2Numerical}\left[-c+d,\ \frac{a+b\,c}{\sqrt{1+b^2}},\ -\frac{b}{\sqrt{1+b^2}}\right]\right)$$

In[43]:=
```
w1[16] /. {S0 → St, KS → KS, K0 → KO, ϵc → ϵCut} /.
  𝓘[p1_, p2_, p3_, p4_] → 𝓘Numerical[p1, p2, p3, p4] /. 𝒩[a_] → 𝒩Numerical[a]
```
Out[43]=
$$\mathrm{COC} = -\frac{1}{2}\,\mathrm{KO}\,\mathrm{Erfc}\left[\frac{\epsilon\mathrm{Cut}}{\sqrt{2}}\right] + e^{-\frac{b^2}{2}}\,\mathrm{St}\,\mathcal{I}\mathrm{Numerical}[a, b, \mathrm{b}, \epsilon\mathrm{Cut}] - \mathrm{KS}\,\mathcal{I}\mathrm{Numerical}\left[a - \sigma\sqrt{\tau}, b, 0, \epsilon\mathrm{Cut}\right]$$

With these expressions in hand, I implement the valuation function COCSemiClosed. Note here that I do not retype the expressions such as $\mathcal{I}$Numerical, I cut and paste them from the derived expressions above, thus lessening the possibility of typographical errors.

In[44]:=
```
Clear[𝒩2Numerical, ℐNumerical, COCSemiClosed];
𝒩2Numerical[a_, b_, ρ_] :=
  NIntegrate[PDF[BinormalDistribution[ρ], {x, y}], {x, -∞, a}, {y, -∞, b}];
ℐNumerical[a_, b_, c_, d_] :=
  E^(c^2/2) (-𝒩2Numerical[-c + d, (a + b c)/Sqrt[1 + b^2], -b/Sqrt[1 + b^2]] + 𝒩Numerical[(a + b c)/Sqrt[1 + b^2]]);
COCSemiClosed[St_, σ_, TO_, KO_, TS_, KS_] :=
 Module[{r = 0, q = 0, τ, b, 𝒶, 𝒷, ϵCut, SO, COC},
  τ = -TO + TS;
  b = Sqrt[TO] σ;
  𝒶 = -b^2/(2 σ Sqrt[τ]) + (σ Sqrt[τ])/2 - Log[KS]/(σ Sqrt[τ]) + Log[St]/(σ Sqrt[τ]);
  𝒷 = b/(σ Sqrt[τ]);
  SO[ϵ_] := St Exp[(r - q - σ^2/2) τ + ϵ σ Sqrt[τ]];
  ϵCut = FindRoot[EuroCall[SO[x], KS, 0, 0, σ, τ] == KO, {x, 0}][[1, 2]];
  COC = E^(-b^2/2) St ℐNumerical[𝒶, 𝒷, b, ϵCut] -
    KS ℐNumerical[𝒶 - σ Sqrt[τ], 𝒷, 0, ϵCut] - KO 𝒩Numerical[-ϵCut]]
```

I perform numerical calculations with this function in Section 3.

2 Evaluation of $\mathcal{I}$(a, b, c, d)

The integral $\mathcal{I}$(a, b, c, d) will be shown to involve the cumulative binormal distribution

$$\mathcal{N}_2(a,\ b,\ \rho) = \int_{-\infty}^{a}\int_{-\infty}^{b} n_2(x,\ y,\ \rho)\, dy\, dx\ =\ \mathbb{P}(x\ <\ a,\ y\ <\ b)$$

where

$$n_2(x, y,\ \rho)\ =\ \frac{1}{2\,\pi\,\sqrt{1-\rho^2}}\, e^{-\frac{1}{2}\frac{x^2-2\,x\,y\,\rho+y^2}{(1-\rho^2)}}$$

Mathematica has a closed form for n_2 but not for $\mathcal{N}_2$

In[48]:=
```
With[{dist = BinormalDistribution[ρ]},
 {PDF[dist, {x, y}], CDF[dist, {x, y}]}]
```

Out[48]=

$$\left\{\frac{e^{-\frac{x^2+y^2-2\,x\,y\,\rho}{2\,(1-\rho^2)}}}{2\,\pi\,\sqrt{1-\rho^2}},\ \text{CDF}[\text{BinormalDistribution}[\rho],\ \{x,\ y\}]\right\}$$

I now proceed with a Mathematica evaluation of $\mathcal{I}$[a, b, c, d]

$$\mathcal{I}(a,\ b,\ c,\ d) \equiv \int_d^\infty d\epsilon\, e^{c\,\epsilon}\, n(\epsilon)\, \mathcal{N}(a\ +\ b\,\epsilon)$$

After many operational steps I will find

$$\mathcal{I}(a,\ b,\ c,\ d) \equiv \int_d^\infty d\epsilon\, e^{c\,\epsilon}\, n(\epsilon)\, \mathcal{N}(a\ +\ b\,\epsilon) = e^{\frac{c^2}{2}}\left(\mathcal{N}\left[\frac{a+b\,c}{\sqrt{1+b^2}}\right] - \mathcal{N}_2\left[-c+d,\ \frac{a+b\,c}{\sqrt{1+b^2}},\ -\frac{b}{\sqrt{1+b^2}}\right]\right)$$

Start with

In[49]:=
```
w2[1] = Int[Exp[c ϵ] n[ϵ] N[b ϵ + a], {ϵ, d, ∞}]
```

Out[49]=

$$\text{Int}\left[e^{c\,\epsilon}\, n[\epsilon]\, \mathcal{N}[a+b\,\epsilon],\ \{\epsilon,\ d,\ \infty\}\right]$$

Introduce an integral form form for $\mathcal{N}$[a + b ϵ]. First, consider the integral representation

In[50]:=
```
w2[2] = N[a + b ϵ] == Int[n[η], {η, -∞, a + b ϵ}]
```

Out[50]=

$$\mathcal{N}[a+b\,\epsilon] == \text{Int}[n[\eta],\ \{\eta,\ -\infty,\ a+b\,\epsilon\}]$$

To remove the ϵ dependence from the upper limit of integration, make the change of variables.

In[51]:=
```
w2[3] = ζ == η - b ϵ
```

Out[51]=

$$\zeta == -b\,\epsilon + \eta$$

Then dζ = dη and the limits of integration become η = -∞ → ζ = -∞ and η = a + b ϵ → ζ = a

In[52]:=
```
w2[4] = w2[2] /. {η, -∞, a + b ϵ} → {ζ, -∞, a} /. Sol[w2[3], η]
```

Out[52]=

$$\mathcal{N}[a+b\,\epsilon] == \text{Int}[n[b\,\epsilon+\zeta],\ \{\zeta,\ -\infty,\ a\}]$$

and

In[53]:=
```
w2[5] = w2[1] /. (w2[4] // ER)
```

Out[53]=

$$\text{Int}\left[e^{c\,\epsilon}\,\text{Int}[n[b\,\epsilon+\zeta],\ \{\zeta,\ -\infty,\ a\}]\, n[\epsilon],\ \{\epsilon,\ d,\ \infty\}\right]$$

Write this as a double integral

In[54]:=
```
w2[6] = w2[5] /. Int[a_ Int[b_, lim2_], lim1_] → Int2[a b, lim1, lim2]
```

Out[54]=

$$\text{Int2}\left[e^{c\,\epsilon}\, n[\epsilon]\, n[b\,\epsilon + \zeta],\ \{\epsilon, d, \infty\},\ \{\zeta, -\infty, a\}\right]$$

Introduce explicit expressions for the terms involving n

In[55]:=
```
w2[7] = w2[6] /. n[x_] → PDF[NormalDistribution[], x] /. {ϵ → x, ζ → y}
```

Out[55]=

$$\text{Int2}\left[\frac{e^{c\,x-\frac{x^2}{2}-\frac{1}{2}(b\,x+y)^2}}{2\,\pi},\ \{x, d, \infty\},\ \{y, -\infty, a\}\right]$$

The next task is to manipulate this expression into a form that can be matched with to n_2

$$n_2(X,\ Y,\ \rho) = \frac{e^{-\frac{X^2-2\rho X Y+Y^2}{2(1-\rho^2)}}}{2\,\pi\sqrt{1-\rho^2}}$$

The general form of the bivariate normal distribution is

In[56]:=
```
w2[8] = κ PDF[BinormalDistribution[{μx, μy}, {σx, σy}, ρ], {x, y}] dx dy
```

Out[56]=

$$\left(dx\,dy\,e^{-\frac{\frac{(x-\mu x)^2}{\sigma x^2}+\frac{(y-\mu y)^2}{\sigma y^2}-\frac{2(x-\mu x)(y-\mu y)\rho}{\sigma x\,\sigma y}}{2(1-\rho^2)}}\,\kappa\right) \Big/ \left(2\,\pi\sqrt{1-\rho^2}\,\sigma x\,\sigma y\right)$$

where κ is a normalization constant to be determined. The expanded integrand of w2[7] is

In[57]:=

$$\text{w2[9]} = \frac{e^{c\,x-\frac{x^2}{2}-\frac{1}{2}(b\,x+y)^2}}{2\,\pi}\ \text{// ExpandAll}$$

Out[57]=

$$\frac{e^{c\,x-\frac{x^2}{2}-\frac{b^2 x^2}{2}-b\,x\,y-\frac{y^2}{2}}}{2\,\pi}$$

I equate the arguments of the exponential terms for these two expressions

In[58]:=
```
w2[10] = (w2[8] /. a_ Exp[b_] → b) == (w2[9] /. a_ Exp[b_] → b) // ExpandAll
```

Out[58]=

$$-\frac{x^2}{(2-2\rho^2)\,\sigma x^2}+\frac{2\,x\,\mu x}{(2-2\rho^2)\,\sigma x^2}-\frac{\mu x^2}{(2-2\rho^2)\,\sigma x^2}-\frac{y^2}{(2-2\rho^2)\,\sigma y^2}+\frac{2\,y\,\mu y}{(2-2\rho^2)\,\sigma y^2}-\frac{\mu y^2}{(2-2\rho^2)\,\sigma y^2}+\frac{2\,x\,y\,\rho}{(2-2\rho^2)\,\sigma x\,\sigma y}-\frac{2\,y\,\mu x\,\rho}{(2-2\rho^2)\,\sigma x\,\sigma y}-\frac{2\,x\,\mu y\,\rho}{(2-2\rho^2)\,\sigma x\,\sigma y}+\frac{2\,\mu x\,\mu y\,\rho}{(2-2\rho^2)\,\sigma x\,\sigma y} = c\,x-\frac{x^2}{2}-\frac{b^2 x^2}{2}-b\,x\,y-\frac{y^2}{2}$$

There are 5 parameters to be determined

In[59]:=
```
w2[11] = {MapEqn[Coefficient[#, x^2] &, w2[10]],
  MapEqn[Coefficient[#, y^2] &, w2[10]], MapEqn[Coefficient[#, x y] &, w2[10]],
  MapEqn[Coefficient[#, x] &, w2[10]] /. y → 0,
  MapEqn[Coefficient[#, y] &, w2[10]] /. x → 0}
```

Out[59]=
$$\left\{-\frac{1}{(2-2\rho^2)\,\sigma x^2} == -\frac{1}{2}-\frac{b^2}{2},\ -\frac{1}{(2-2\rho^2)\,\sigma y^2} == -\frac{1}{2},\ \frac{2\rho}{(2-2\rho^2)\,\sigma x\,\sigma y} == -b,\ \frac{2\mu x}{(2-2\rho^2)\,\sigma x^2}-\frac{2\mu y\,\rho}{(2-2\rho^2)\,\sigma x\,\sigma y} == c,\ \frac{2\mu y}{(2-2\rho^2)\,\sigma y^2}-\frac{2\mu x\,\rho}{(2-2\rho^2)\,\sigma x\,\sigma y} == 0\right\}$$

In[60]:=
```
w2[12] = MapEqn[Simplify, w2[11]]
```

Out[60]=
$$\left\{\frac{1}{2(-1+\rho^2)\,\sigma x^2} == \frac{1}{2}(-1-b^2),\ \frac{1}{2(-1+\rho^2)\,\sigma y^2} == -\frac{1}{2},\ \frac{\rho}{\sigma x\,\sigma y-\rho^2\,\sigma x\,\sigma y} == -b,\ \frac{\mu y\,\rho\,\sigma x-\mu x\,\sigma y}{(-1+\rho^2)\,\sigma x^2\,\sigma y} == c,\ \frac{-\mu y\,\sigma x+\mu x\,\rho\,\sigma y}{(-1+\rho^2)\,\sigma x\,\sigma y^2} == 0\right\}$$

In[61]:=
```
w2[13] = Solve[w2[12], {σx, σy, ρ, μx, μy}] // Simplify
```

Out[61]=
$$\left\{\left\{\sigma x\to -1,\ \sigma y\to -\sqrt{1+b^2},\ \rho\to -\frac{b}{\sqrt{1+b^2}},\ \mu x\to c,\ \mu y\to -b\,c\right\},\right.$$
$$\left\{\sigma x\to -1,\ \sigma y\to \sqrt{1+b^2},\ \rho\to \frac{b}{\sqrt{1+b^2}},\ \mu x\to c,\ \mu y\to -b\,c\right\},$$
$$\left\{\sigma x\to 1,\ \sigma y\to \sqrt{1+b^2},\ \rho\to -\frac{b}{\sqrt{1+b^2}},\ \mu x\to c,\ \mu y\to -b\,c\right\},$$
$$\left.\left\{\sigma x\to 1,\ \sigma y\to -\sqrt{1+b^2},\ \rho\to \frac{b}{\sqrt{1+b^2}},\ \mu x\to c,\ \mu y\to -b\,c\right\}\right\}$$

The financially appropriate branch is the one with σx and σy positive

In[62]:=
```
w2[14] = w2[13][[3]]
```

Out[62]=
$$\left\{\sigma x\to 1,\ \sigma y\to \sqrt{1+b^2},\ \rho\to -\frac{b}{\sqrt{1+b^2}},\ \mu x\to c,\ \mu y\to -b\,c\right\}$$

Thus, the integrand of w2[7]

In[63]:=
```
w2[7]
```

Out[63]=
$$\text{Int2}\left[\frac{e^{c\,x-\frac{x^2}{2}-\frac{1}{2}(b\,x+y)^2}}{2\pi},\ \{x, d, \infty\},\ \{y, -\infty, a\}\right]$$

is equivalent to $n_2(x, y, -\frac{b}{\sqrt{1+b^2}})$. It still remains to rescale the integration variables X = $\frac{x-\mu x}{\sigma x}$, Y = $\frac{y-\mu y}{\sigma y}$, and to account for the fact that n_2 is normalized to unity while w2[7] is not. To calculate the normalization set x = y = 0

In[64]:=
```
w2[15] = w2[8] == w2[9] /. dx → 1 /. dy → 1 /. x → 0 /. y → 0
```

Out[64]=

$$\frac{e^{-\frac{\frac{\mu x^2}{\sigma x^2}+\frac{\mu y^2}{\sigma y^2}-\frac{2\,\mu x\,\mu y\,\rho}{\sigma x\,\sigma y}}{2\,(1-\rho^2)}}\,\kappa}{2\,\pi\,\sqrt{1-\rho^2}\,\sigma x\,\sigma y} == \frac{1}{2\,\pi}$$

In[65]:=
```
w2[16] = Solve[w2[15], κ][[1, 1]]
```

Out[65]=

$$\kappa \to e^{\frac{\mu x^2}{2\,(1-\rho^2)\,\sigma x^2}+\frac{\mu y^2}{2\,(1-\rho^2)\,\sigma y^2}-\frac{\mu x\,\mu y\,\rho}{(1-\rho^2)\,\sigma x\,\sigma y}}\,\sqrt{1-\rho^2}\,\sigma x\,\sigma y$$

In[66]:=
```
w2[17] = w2[16] /. w2[14] // Simplify // PowerExpand
```

Out[66]=

$$\kappa \to e^{\frac{c^2}{2}}$$

I am now prepared to rewrite the integral w2[7] in a form that is *almost* consistent with the definition of the cumulative bivariate distribution in the “standard” form.

In[67]:=

$$\text{w2[18] = w2[7] /. } \frac{e^{c\,x-\frac{x^2}{2}-\frac{1}{2}(b\,x+y)^2}}{2\,\pi} \to \kappa\, n_2(x, y, \rho) \text{ /. d} \to (d-\mu x)/\sigma x \text{ /. a} \to (a-\mu y)/\sigma y \text{ /. w2[17]}$$

Out[67]=

$$\text{Int2}\left[e^{\frac{c^2}{2}}\, n_2[x, y, \rho], \left\{x, \frac{d-\mu x}{\sigma x}, \infty\right\}, \left\{y, -\infty, \frac{a-\mu y}{\sigma y}\right\}\right]$$

Using the explicit mapping parameters

In[68]:=
```
w2[19] = w2[18] /. w2[14] /. Int2[a_ b_, lim1_, lim2_] /;
    And[FreeQ[a, x, ∞], FreeQ[a, y, ∞]] → a Int2[b, lim1, lim2]
```

Out[68]=

$$e^{\frac{c^2}{2}}\,\text{Int2}\left[n_2\left[x, y, -\frac{b}{\sqrt{1+b^2}}\right], \{x, -c+d, \infty\}, \left\{y, -\infty, \frac{a+b\,c}{\sqrt{1+b^2}}\right\}\right]$$

Note that the limits of integration in w2[19] are not in the standard form for the cumulative standard bivariate distribution. In Appendix A, I calculate the identity

$$\int_{-\infty}^{a} dx \int_{b}^{\infty} dy\, n_2(x, y, \rho) = \mathcal{N}[a] - \mathcal{N}_2[a, b, \rho]$$

and use that to obtain the standard form

In[69]:=
```
w2["final result"] = I[a, b, c, d] == w2[19] /.
   co_ Int2[n2[x, y, ρ_], {x, A_, ∞}, {y, -∞, B_}] → co (N[B] - N2[A, B, ρ])
```

Out[69]=

$$\mathcal{I}[a, b, c, d] == e^{\frac{c^2}{2}}\left(\mathcal{N}\left[\frac{a+b\,c}{\sqrt{1+b^2}}\right] - \mathcal{N}_2\left[-c+d, \frac{a+b\,c}{\sqrt{1+b^2}}, -\frac{b}{\sqrt{1+b^2}}\right]\right)$$

After going through such a series of symbolic manipulations, it is worthwhile to perform a numerical check of beginning, intermediate, and final forms to establish confidence in the derived forms.

In[70]:=
```
Module[{n, N, a = 1, b = 1, c = 2 , d = 2, ρ, κ, n2, N2},
 ρ = - b/Sqrt[1 + b^2];
 κ = ⅇ^(c^2/2);
 n[ϵ_] := PDF[NormalDistribution[], ϵ];
 N[x_] := CDF[NormalDistribution[], x];
 n2[x_, y_, ρ_] := PDF[BinormalDistribution[ρ], {x, y}];
 N2[a_, b_, ρ_] := NIntegrate[n2[x, y, ρ], {x, -∞ , a}, {y, -∞ , b}];
 {NIntegrate[Exp[c ϵ] n[ϵ] N[a + b ϵ], {ϵ, d, ∞}],
  NIntegrate[ⅇ^(c x - x^2/2 - 1/2 (b x + y)^2)/(2 π), {x, d, ∞}, {y, -∞, a}],
  NIntegrate[κ n2[X, Y, ρ], {X, -c + d, ∞}, {Y, -∞, a}],
  ⅇ^(c^2/2) (N[(a + b c)/Sqrt[1 + b^2]] - N2[-c + d, (a + b c)/Sqrt[1 + b^2], - b/Sqrt[1 + b^2]])}]
```

Out[70]=
```
{3.69347, 3.69347, 3.60153, 3.69347}
```

Increasing the accuracy requirement beyond the default value for NIntegrate on the third term would bring it in line.

3 Numerical calculations

I generate some numerical results for the Call on Call compound options, by implementing three different models

Model 1 - Brute force, single variable numerical integration of the defining expression eqn(6) - *COCNumericalSQ*
Model 2 - Single quadrature numerical integration of the defining expression eqn(13) with a numerically determine lower cutoff - *COCNumericalSQCutoff*
Model 3 - Implementation of the semi-closed form eqn(7) equivalent to Geske's 1978 result - *COCSemiClosed*.

In[71]:=
```
Clear[nNumerical, NNumerical, EuroCall, COCNumericalSQ, COCNumericalSQCutoff];
nNumerical[ϵ_] := PDF[NormalDistribution[], ϵ];
NNumerical[a_] := CDF[NormalDistribution[], a];
EuroCall[St_, K_, r_, q_, σ_, τ_] :=
  Module[{d},
   d = (Log[St / K] + (r - q + σ^2/2) τ) / (σ Sqrt[τ]);
   Exp[-q τ] St NNumerical[d] - Exp[-r τ] K NNumerical[-σ Sqrt[τ] + d]];

(* Model 1 *)
COCNumericalSQ[St_, r_, q_, σ_, TO_, KO_, TS_, KS_] :=
  Module[{d, SO},
   SO[ϵ_] := St Exp[(r - q - σ^2/2) TO + ϵ σ Sqrt[TO]];
   Exp[-r TO] NIntegrate[
     nNumerical[ϵ] Max[EuroCall[SO[ϵ], KS, r, q, σ, TS - TO] - KO, 0], {ϵ, -∞, ∞}]];

COCNumericalSQCutoff[St_, r_, q_, σ_, TO_, KO_, TS_, KS_] :=
 Module[{d, SO, ϵCut},
  SO[ϵ_] := St Exp[(r - q - σ^2/2) TO + ϵ σ Sqrt[TO]];
  (* determine the lower limit cutoff *)
  ϵCut = FindRoot[EuroCall[SO[x], KS, r, q, σ, TS - TO] == KO, {x, 0}][[1, 2]];
  Exp[-r TO] NIntegrate[
    nNumerical[ϵ] Max[EuroCall[SO[ϵ], KS, r, q, σ, TS - TO] - KO, 0], {ϵ, ϵCut, ∞}]]
```

The model COCSemiClosed was implemented at the end of Section 2.

I compare the three models for representative parameters

Consider a 1 month call-on-call option struck at 3 on a 2 month call option on a stock struck at 100. The market conditions are St = 100, r = 0, q = 0, σ = 0.2.

In[77]:=
```
Module[{St = 100, r = 0.0, q = 0.0, σ = 0.2,
  TO = 1/12., KS = 100, TS = 3/12., KO = 3, info, infoParams, G},
 info = {{"Model 1", "Model 2", "Model 3"},
   {COCNumericalSQ[St, r, q, σ, TO, KO, TS, KS], COCNumericalSQCutoff[St,
     r, q, σ, TO, KO, TS, KS], COCSemiClosed[St, σ, TO, KO, TS, KS]}};
 infoParams = {{"St", "r", "q", "σ", "KS", "TS", "KO", "TO"},
   {NF2@St, r, q, NF2@σ, NF2@KS, NF4@TS, NF2@KO, NF4@TO}};
 G[1] = LGrid[info, "Model comparison for representative parameters"];
 G[2] = LGrid[infoParams, "Parameters values"];
 Grid[{{G[1]}, {G[2]}}]]
```

Out[77]=

Model comparison for representative parameters

Model 1	Model 2	Model 3
1.68439	1.68439	1.68358

Parameters values

St	r	q	σ	KS	TS	KO	TO
100.00	0.	0.	0.20	100.00	0.2500	3.00	0.0833

I plot some comparative results for Model 2 (single quadrature with cutoff) and Model 3 (semi-closed form). Using the predetermined lower cutoff for the the numerical integration is more efficient that allowing Mathematica's adaptive sampling algorithm to determine it automatically.

In[78]:=

```
Module[{St = 100, r = 0.0, q = 0.0, σ = 0.2,
  TO = 1/12., KS = 100, TS = 3/12., KO = 3, values, info, G},
 values["cutoff"] = Table[{KS, COCNumericalSQCutoff[St, r, q, σ, TO, KO, TS, KS]},
   {KS, 0.9 St, 1.1 St, 0.01 St}];
 values["semi-closed"] = Table[{KS, COCSemiClosed[St, σ, TO, KO, TS, KS]},
   {KS, 0.9 St, 1.1 St, 0.01 St}];
 info = {{"St", "r", "q", "σ", "TS", "KO", "TO"},
   {NF2@St, r, q, NF2@σ, NF4@TS, NF2@KO, NF4@TO}};
 G[1] = ListPlot[values["cutoff"], Joined → True,
   Epilog → {OC[#, Black] & /@ values["semi-closed"]},
   AxesLabel → {Stl["KS"], Stl["COC"]}, PlotLabel →
    "COC as function of stock option strike KS\nModel2 - Blue line, Model 3
      - Black circles", ImageSize → {450, 250}];
 G[2] = LGrid[info, "Parameters"];
 Grid[{{G[1]}, {G[2]}}]]
```

Out[78]=



Parameters

St	r	q	σ	TS	KO	TO
100.00	0.	0.	0.20	0.2500	3.00	0.0833

In[79]:=

```
Module[{St = 100, r = 0.0, q = 0.0, σ = 0.2,
  TO = 1/12., KS = 100, TS = 3/12., KO = 3, values, info, G},
 values["cutoff"] = Table[{KO, COCNumericalSQCutoff[St, r, q, σ, TO, KO, TS, KS]},
   {KO, 2, 4, 0.25}];
 values["semi-closed"] = Table[{KO, COCSemiClosed[St, σ, TO, KO, TS, KS]},
   {KO, 2, 4, 0.25}];
 info = {{"St", "r", "q", "σ", "KS", "TS", "TO"},
   {NF2@St, r, q, NF2@σ, NF2@KS, NF4@TS, NF4@TO}};
 G[1] = ListPlot[values["cutoff"], Joined → True,
   Epilog → {OC[#, Black] & /@ values["semi-closed"]},
   AxesLabel → {Stl["K_O"], Stl["COC"]}, PlotLabel →
    "COC as function of option on option strike KO\nModel2 - Blue line, Model
      3 - Black circles", ImageSize → {450, 250}];
 G[2] = LGrid[info, "Parameters"];
 Grid[{{G[1]}, {G[2]}}]]
```

Out[79]=



Parameters

St	r	q	σ	KS	TS	TO
100.00	0.	0.	0.20	100.00	0.2500	0.0833

I note that Mathematica’s “FinancialDerivatives” package also contains models for compound options. I compare the model valuations developed in this notebook with the relevant FinancialDerivatives’ {“CompoundCall”,”European”,”Call”} model. The agreement is close for the representative case but I did note discrepancies for other parameter values. By close, I do not mean identical to x decimal places. Rather, close means that discrepancies are within a typical market maker generated bid-ask spread.

In[80]:=

```
Module[{St = 100, r = 0.0, q = 0.0, σ = 0.2,
  TO = 1/12., KS = 100, TS = 3/12., KO = 3, info, infoParams, G},
 info = {{"Model 1", "Model 2", "Model 3", "FinancialDerivatives"},
   {COCNumericalSQ[St, r, q, σ, TO, KO, TS, KS], COCNumericalSQCutoff[
     St, r, q, σ, TO, KO, TS, KS], COCSemiClosed[St, σ, TO, KO, TS, KS],
    FinancialDerivative[{"CompoundCall", "European", "Call"},
     {"StrikePriceUnderlying" → KS, "StrikePrice" → KO,
      "ExpirationUnderlying" → TS , "Expiration" → TO},
     {"CurrentPrice" → St, "Dividend" → q, "Volatility" → σ, "InterestRate" → r}]}};
 infoParams = {{"St", "r", "q", "σ", "KS", "TS", "KO", "TO"},
   {NF2@St, r, q, NF2@σ, NF2@KS, NF4@TS, NF2@KO, NF4@TO}};
 G[1] = LGrid[info, "Model comparison for representative parameters"];
 G[2] = LGrid[infoParams, "Parameters values"];
 Grid[{{G[1]}, {G[2]}}]]
```

Out[80]=

Model comparison for representative parameters

Model 1	Model 2	Model 3	FinancialDerivatives
1.68439	1.68439	1.68358	1.68407

Parameters values

St	r	q	σ	KS	TS	KO	TO
100.00	0.	0.	0.20	100.00	0.2500	3.00	0.0833

Close agreement in financial derivative practice does not mean identical to x decimal places. Rather, close means that discrepancies are within a typical market maker generated bid-ask spread. The main reason for this lack of precision is that the values of the input market parameters -- S, r, σ are constantly changing in at least the third decimal place. Further, the various assumptions that underlie the no-arbitrage option theory are not satisfied to high accuracy. For example, the theory presupposes a perfectly liquid option market and the ability of market participants to take instantaneous advantage of arbitrages — doesn't happen in the real world.

An unfortunate weakness of Mathematica's FinancialDerivatives package is that the documentation is quite sparse and is insufficient to allow quantitative discrepancies between alternative models to be resolved. The documentation refers to "typical" parameters and contract specifications. This is a bit annoying since, if there is one thing I learned in ten plus years of experience as a quant for two large investment banks, it is that there are no standard definitions for derivative financial instruments. Derivative traders develop argots and it sometimes requires a translator (usually one of the desk quants) for risk managers or higher management to be able to communicate effectively with them. There is often disagreement between different trading groups (even within the same firm!) as to the definition of a derivative instrument and especially as to how that instrument should be valued. Instruments traded with an external counterparty require a "term sheet" - a detailed legalistic sounding document specifying terms and conditions. Also, contemporary government regulatory policies requires extensive documentation and testing. There are good paying jobs with titles like "model validator", "model risk specialist", or "model governance specialist."

Finally, I mention that it should be appreciated that fair values for the COC calculated using the models above should NOT be depended on when purchasing the option in an over-the-counter financial market. A market maker who would sell this option to you would use a more sophisticated dynamical model for the stock price, one carefully calibrated to present market conditions. Such a model would account for term structure of interest rates (the r term is more complicated), considerations about actual dividends paid by the underlying corporation issuing the stock (the q term is more complicated) and, most important, the model would account for the observable fact that volatility is neither constant or uniform over all strikes and option tenors (the σ term is much more complicated). The practical way for a non expert party to buy a compound option would be to shop around and obtain asking prices from several trading desks.

Nonetheless, it is useful for quants to know how to value such options under GBM. Such tractable models provide intuition and often serve as the basis for testing more sophisticated models. Simplified models are also useful in enterprise risk management systems where many tens of thousands of instruments are being valued and hedged on a daily basis. The incorporation of longer running complex computational models in those systems is not technically feasible.

The GBM based approach to the COC option would be useful in “real option” applications where option theoretics are used to inform decisions based on future uncertainties and contingencies. In such applications, sharp-eyed, model-rich market makers with asymmetric information would not be on the other side of the trade.

Appendix A - A required identity

In the calculation of $\mathcal{I}$(a, b, c, d) the following identity is required.

$$\int_a^{\infty} \mathbf{dx} \int_{-\infty}^{b} \mathbf{dx}\, n_2(x,\ y,\ \rho) = \mathcal{N}(b) - \mathcal{N}2(a, b, \rho)$$

Rather than proceed step by step in the manner above, I take a different approach that streamlined the calculation for this simpler special case. I again define a struct to represent the integral but call it Jnt this time to avoid confusion with the Int form used above. I assign properties to the function Jit that facilitate its valuation.

In[81]:=
```
Clear[Jnt, n2];
n2[x_, y_, ρ_] := PDF[BinormalDistribution[ρ], {x, y}];
(* Distribute across addition *)
Jnt[a_ + b_, c_] := Jnt[a, c] + Jnt[b, c];
(* Remove constant terms *)
Jnt[a_ b_, {var_, lo_, hi_}] /; FreeQ[a, var, ∞] := a Jnt[b, {var, lo, hi}];
(* Expand ∫_a^∞ f[x] ⅆx →
 ∫_-∞^∞ f[x] ⅆx - ∫_-∞^a f[x] ⅆx    to obtain form consistent with defintion of 𝒩₂ *)
Jnt[arg_, {var_, lo_, ∞}] /; Not[SameQ[lo, -∞]] :=
  Jnt[arg, {var, -∞, ∞}] - Jnt[arg, {var, -∞, lo}];
(* Interchange order of integration so that
 ∫_-∞^∞ n₂(x, y, ρ) dx or ∫_-∞^∞ n₂(x, y, ρ) dy will evaluate first. *)
Jnt[Jnt[n2[x, y, ρ], {var1_, -∞, hi_}], {var2_, -∞, ∞}] /; Not[SameQ[hi, -∞]] :=
  Jnt[Jnt[n2[x, y, ρ], {var2, -∞, ∞}], {var1, -∞, hi}];
(* Match definition of 𝒩₂ *)
Jnt[Jnt[n2[x, y, ρ], {x, -∞, a}], {y, -∞, b}] := 𝒩2[a, b, ρ];
(* Evaluates ∫_-∞^∞ n₂(x, y, ρ) ⅆx or ∫_-∞^∞ n₂(x, y, ρ) ⅆy *)
Jnt[arg_, {var_, -∞, ∞}] := Module[{w},
   w[1] = Integrate[Activate[arg, n2], {var, -∞, ∞}, Assumptions → {Re[ρ^2] ≤ 1}]];
(* Evaluates ∫_-∞^∞ n₁(x) ⅆx or ∫_-∞^∞ n₁(y) ⅆy *)
Jnt[arg_, {var_, -∞, hi_}] /; FreeQ[arg, n2, ∞] :=
  Integrate[arg, {var, -∞, hi}] /. ErfRules;
```

A sequence of intermediate evaluations occur and rules fire, but the final result is obtained by simply evaluating the defined expression.

In[90]:=
```
wA[1] = Jnt[Jnt[Inactivate[n2[x, y, ρ], n2], {x, a, ∞}], {y, -∞, b}]
```

Out[90]=
```
𝒩[b] - 𝒩2[a, b, ρ]
```

Appendix B Visualization

Figure

In[91]:=

```
Module[{t = 0, TO = 1, TS = 2, ymax = 1},
 Graphics[{{Directive[Thick, Black], Line[{{t, 0}, {TS, 0}}]},
   Line[{{0, 0}, {0, ymax}}], {Directive[Black, Dashed],
    Line[{{TS, 0}, {TS, ymax}}], Line[{{TO, 0}, {TO, ymax}}]},
   Text[Style["t", Bold], {0, -0.12}],
   Text[Style["T_O", Bold], {1, -0.12}],
   Text[Style["T_S", Bold], {2, -0.12}],
   Text[Style["Option on Stock expires at T_S", Bold], {1.1, 0.5}, {-1, 0}],
   Text[Style["Strike = K_S", Bold], {1.1, 0.35}, {-1, 0}],
   Text[Style["Option on Option expires at T_O", Bold], {0.1, 0.5}, {-1, 0}],
   Text[Style["Strike = K_O", Bold], {0.1, 0.35}, {-1, 0}]}, AspectRatio → 0.25]]
```

Out[91]=



Functions

In[10]:=

```
Clear[𝒩];
𝒩::usage = "Cumulative standard normal distribution function";
𝒩[z_?NumberQ] := (1/2) (1 + Erf[z/√2]) // N;
```

In[13]:=

```
Clear[EuroCall];
EuroCall[S_, K_, r_, q_, σ_, τ_] :=
Module[{d1},
 If[τ <= 0.00001, Return[Max[S - K, 0.0]]];
 d1 = (Log[S/K] + (r - q + σ^2/2) τ) / (σ Sqrt[τ]);
 S Exp[-q τ] 𝒩[d1] -
   K Exp[-r τ] 𝒩[d1 - σ Sqrt[ τ]]];
```